# Copyright (c) 2011 Nokia Corporation.

QT += core gui opengl

TARGET = qtopenglspritebatchtest
TEMPLATE = app
VERSION = 1.1.1

INCLUDEPATH += .
INCLUDEPATH += ./src_qtopenglspritebatch
INCLUDEPATH += src_general
INCLUDEPATH += src_qtopenglspritebatchtest

SOURCES += \
    src_general/SpriteBatch.cpp \
    src_qtopenglspritebatch/QtOpenGLSpriteBatch.cpp \
    src_qtopenglspritebatchtest/mygamewindow.cpp \
    src_qtopenglspritebatchtest/main.cpp \
    SpriteBatchTester.cpp

HEADERS += \
    src_general/SpriteBatch.h \
    src_qtopenglspritebatch/QtOpenGLSpriteBatch.h \
    src_qtopenglspritebatchtest/mygamewindow.h \
    src_qtopenglspritebatchtest/gamewindow.h \
    SpriteBatchTester.h


symbian {
    # FOR ENABLING HARDWARE FLOATS
    MMP_RULES += "OPTION gcce -march=armv6"
    MMP_RULES += "OPTION gcce -mfpu=vfp"
    MMP_RULES += "OPTION gcce -mfloat-abi=softfp"
    MMP_RULES += "OPTION gcce -marm"

    TARGET.UID3 = 0xee69ded2
    TARGET.EPOCSTACKSIZE = 0x14000
    TARGET.EPOCHEAPSIZE = 0x020000 0x800000
}


# Unix based platforms
unix:!symbian {
    # Common
    BINDIR = /opt/usr/bin
    DATADIR = /usr/share

    target.path = $$BINDIR
    icon64.files += $${GE_PATH}/icons/qtgameenabler.png
    icon64.path = $$DATADIR/icons/hicolor/64x64/apps

    maemo5 {
        # Maemo 5 specific
        desktopfile.files += qtc_packaging/debian_fremantle/$${TARGET}.desktop
        desktopfile.path = $$DATADIR/applications/hildon

        INSTALLS += \
            desktopfile \
            icon64
    }
    else {
        contains(DEFINES, MEEGO_EDITION_HARMATTAN) {
            # Harmattan specific
            desktopfile.files += qtc_packaging/debian_harmattan/$${TARGET}.desktop
            desktopfile.path = $$DATADIR/applications

            INSTALLS += \
                desktopfile \
                icon64
        }
        else {
            # Unix based desktop specific
            target.path = /usr/local/bin
        }
    }

    # Common cont.
    INSTALLS += target
}


windows: {
    QT += multimedia

    message("The project contains desktop ogl")
    message($$INCLUDEPATH)
    message($$LIBS)

}

# End of file.

RESOURCES += \
    resources.qrc











